



Optimization in Engineering Design

Robert J. Vanderbei

2008 Oct 10

Mechanical & Aerospace Engineering
Princeton University

<http://www.princeton.edu/~rvdb>

Engineering = Optimization

Make something.

Make it better.

Make it better yet.

Make it still better.

⋮

Types of Optimization Problems

Constrained vs. unconstrained.

Convex vs. nonconvex.

Smooth vs. nonsmooth.

Continuous vs. discrete (integer).

Local vs. global.

My main interest: finding *locally optimal* solutions to *nonconvex, constrained, smooth* optimization problems in which all variables are assumed to be *continuous* (based on LOQO).

Types of Algorithms

Black-box-around-legacy-code vs. start-from-scratch (Dennis vs. Betts).

My main interest: the *start-from-scratch* approach (based on AMPL).

LOQO: An algorithm for constrained optimization

LOQO solves problems in the following form:

$$\begin{aligned} & \text{minimize} && f(x) \\ & \text{subject to} && b \leq h(x) \leq b + r, \\ & && l \leq x \leq u \end{aligned}$$

The functions $f(x)$ and $h(x)$ must be twice differentiable (at least at points of evaluation).

The standard *interior-point paradigm* is used:

- Add slacks.
- Replace nonnegativities (e.g., $w \geq 0$) with parametrized logarithmic barrier terms $(-\mu \log w)$ in objective function.
- Write first-order optimality conditions.
- Rewrite optimality conditions in primal-dual symmetric form.
- Use Newton's method to get search directions...

Interior-Point Paradigm Continued

- Use Newton's method to get search directions:

$$\begin{bmatrix} -H(x, y) - D & A^T(x) \\ A(x) & E \end{bmatrix} \begin{bmatrix} \Delta x \\ \Delta y \end{bmatrix} = \begin{bmatrix} \nabla f(x) - A^T(x)y \\ -h(x) + \mu Y^{-1}e \end{bmatrix}.$$

Here, D and E are diagonal matrices involving slack variables,

$$H(x, y) = \nabla^2 f(x) - \sum_{i=1}^m y_i \nabla^2 h_i(x) + \lambda I, \text{ and } A(x) = \nabla h(x),$$

where λ is chosen to ensure appropriate descent properties.

- Compute step lengths to ensure positivity of slack variables.
- Shorten steps further to ensure a reduction in either infeasibility or in the barrier function.
- Step to new point, reduce barrier parameter (μ) and repeat.

Telescope Design

for Planet-Finding

Terrestrial Planet Finder Telescope (TPF)

- DETECT: Search 150-500 nearby (5-15 pc distant) Sun-like stars for Earth-like planets.
- CHARACTERIZE: Determine basic physical properties and measure “biomarkers”, indicators of life or conditions suitable to support it.

Why Is It Hard? Can't Hubble do it?

- If the star is Sun-like and the planet is Earth-like, then the reflected visible light from the planet is 10^{-10} times as bright as the star. This is a difference of 25 magnitudes!
- If the star is 10 pc (33 ly) away and the planet is 1 AU from the star, the angular separation is 0.1 arcseconds!
- A point source (i.e. star) produces not a point image but an *Airy pattern* consisting of an *Airy disk* surrounded by a system of *diffraction rings* completely covering the nearby planet.
- By *apodizing* the entrance pupil, one can control the shape and strength of the diffraction rings.

Electric Field

The image-plane *electric field* $E()$ produced by an on-axis plane wave and an apodized aperture defined by an *apodization function* $A()$ is given by

$$E(\xi, \zeta) = \iint_{\bigcirc} e^{i(x\xi+y\zeta)} A(x, y) dy dx$$
$$\vdots$$
$$E(\rho) = 2\pi \int_0^{1/2} J_0(r\rho) A(r) r dr,$$

where J_0 denotes the 0-th order Bessel function of the first kind.

NOTE: The *electric field* depends *linearly* on the *apodization function*.

The unitless pupil-plane “length” r is given as a multiple of the aperture D .

The unitless image-plane “length” ρ is given as a multiple of focal-length times wavelength over aperture ($f\lambda/D$) or, equivalently, as an angular measure on the sky, in which case it is a multiple of just λ/D . (Example: $\lambda = 0.5\mu\text{m}$ and $D = 10\text{m}$ implies $\lambda/D = 10\text{mas}$.)

The *intensity* is the square of the electric field.

Performance Metrics

Inner and Outer Working Angles

$$\rho_{iwa} \quad \rho_{owa}$$

Contrast:

$$E^2(\rho)/E^2(0)$$




Airy Throughput:

$$\frac{\int_0^{\rho_{iwa}} E^2(\rho) 2\pi \rho d\rho}{(\pi(1/2)^2)} = 8 \int_0^{\rho_{iwa}} E^2(\rho) \rho d\rho.$$

Clear Aperture—Airy Pattern

$$\rho_{iwa} = 1.24 \quad \mathcal{T}_{\text{Airy}} = 84.2\% \quad \text{Contrast} = 10^{-2}$$




$$\rho_{iwa} = 748 \quad \mathcal{T}_{\text{Airy}} = 100\% \quad \text{Contrast} = 10^{-10}$$



Clear Aperture—Airy Pattern

$$\rho_{iwa} = 1.24 \quad \mathcal{T}_{\text{Airy}} = 84.2\% \quad \text{Contrast} = 10^{-2}$$

$$\rho_{iwa} = 748 \quad \mathcal{T}_{\text{Airy}} = 100\% \quad \text{Contrast} = 10^{-10}$$



Optimization

Find *apodization* function $A()$ that solves:

$$\begin{aligned} &\text{maximize} && \int_0^{1/2} A(r) 2\pi r dr \\ &\text{subject to} && -10^{-5} E(0) \leq E(\rho) \leq 10^{-5} E(0), && \rho_{\text{iwa}} \leq \rho \leq \rho_{\text{owa}}, \\ &&& 0 \leq A(r) \leq 1, && 0 \leq r \leq 1/2, \end{aligned}$$

Optimization

Find *apodization* function $A()$ that solves:

$$\begin{aligned} &\text{maximize} && \int_0^{1/2} A(r) 2\pi r dr \\ &\text{subject to} && -10^{-5} E(0) \leq E(\rho) \leq 10^{-5} E(0), && \rho_{\text{iwa}} \leq \rho \leq \rho_{\text{owa}}, \\ &&& 0 \leq A(r) \leq 1, && 0 \leq r \leq 1/2, \\ &&& -50 \leq A''(r) \leq 50, && 0 \leq r \leq 1/2 \end{aligned}$$

An infinite dimensional *linear programming* problem.

The AMPL Model

```
function J0;

param pi := 4*atan(1);
param N := 400; # discretization parameter
param rho0 := 4;
param rho1 := 60;

param dr := (1/2)/N;
set Rs ordered := setof {j in 0.5..N-0.5 by 1} (1/2)*j/N;

var A {Rs} >= 0, <= 1, := 1/2;

set Rhos ordered := setof {j in 0..N} j*rho1/N;
set PlanetBand := setof {rho in Rhos: rho>=rho0 && rho<=rho1} rho;

var E0 {rho in Rhos} =
    2*pi*sum {r in Rs} A[r]*J0(2*pi*r*rho)*r*dr;

maximize area: sum {r in Rs} 2*pi*A[r]*r*dr;
subject to sidelobe_pos {rho in PlanetBand}: E0[rho] <= 10(-5)*E0[0];
subject to sidelobe_neg {rho in PlanetBand}: -10(-5)*E0[0] <= E0[rho];





subject to smooth {r in Rs: r != first(Rs) && r != last(Rs)}:
    -50*dr2 <= A[next(r)] - 2*A[r] + A[prev(r)] <= 50*dr2;

solve;
```

The Optimal Apodization

$$\rho_{\text{iwa}} = 4 \quad \mathcal{T}_{\text{Airy}} = 9\%$$

Excellent dark zone. Unmanufacturable.



Concentric Ring Masks

Recall that for circularly symmetric apodizations

$$E(\rho) = 2\pi \int_0^{1/2} J_0(r\rho)A(r)rdr,$$

where J_0 denotes the 0-th order Bessel function of the first kind.

Let

$$A(r) = \begin{cases} 1 & r_{2j} \leq r \leq r_{2j+1}, \quad j = 0, 1, \dots, m-1 \\ 0 & \text{otherwise,} \end{cases}$$

where

$$0 \leq r_0 \leq r_1 \leq \dots \leq r_{2m-1} \leq 1/2.$$

The integral can now be written as a sum of integrals and each of these integrals can be explicitly integrated to get:

$$E(\rho) = \sum_{j=0}^{m-1} \frac{1}{\rho} \left(r_{2j+1} J_1(\rho r_{2j+1}) - r_{2j} J_1(\rho r_{2j}) \right).$$

Mask Optimization Problem

$$\text{maximize } \sum_{j=0}^{m-1} \pi(r_{2j+1}^2 - r_{2j}^2)$$

$$\text{subject to: } -10^{-5}E(0) \leq E(\rho) \leq 10^{-5}E(0), \quad \text{for } \rho_0 \leq \rho \leq \rho_1$$

where $E(\rho)$ is the function of the r_j 's given on the previous slide.

This problem is a semiinfinite nonconvex optimization problem.

The AMPL Model

```
function intrJ0;

param pi := 4*atan(1);
param N := 400; # discretization parameter
param rho0 := 4;
param rho1 := 60;

var r {j in 0..M} >= 0, <= 1/2, := r0[j];

set Rhos2 ordered := setof {j in 0..N} (j+0.5)*rho1/N;
set PlanetBand2 := setof {rho in Rhos2: rho>=rho0 && rho<=rho1} rho;

var E {rho in Rhos2} =
    (1/(2*pi*rho)^2)*sum {j in 0..M by 2}
    (intrJ0(2*pi*rho*r[j+1]) - intrJ0(2*pi*rho*r[j]));

maximize area2: sum {j in 0..M by 2} (pi*r[j+1]^2 - pi*r[j]^2);
subject to sidelobe_pos2 {rho in PlanetBand2}: E[rho] <= 10^(-5)*E[first(rhos2)];
subject to sidelobe_neg2 {rho in PlanetBand2}: -10^(-5)*E[first(rhos2)] <= E[rho];

subject to order {j in 0..M-1}: r[j+1] >= r[j];




solve mask;
```

The Best Concentric Ring Mask

$$\rho_{iwa} = 4 \quad \rho_{owa} = 60$$

$$\mathcal{T}_{\text{Airy}} = 9\%$$

Lay it on glass?



Other Masks

Consider a binary apodization (i.e., a mask) consisting of an opening given by

$$A(x, y) = \begin{cases} 1 & |y| \leq a(x) \\ 0 & \text{else} \end{cases}$$

We only consider masks that are symmetric with respect to both the x and y axes. Hence, the function $a(\cdot)$ is a nonnegative even function.

In such a situation, the electric field $E(\xi, \zeta)$ is given by

$$\begin{aligned} E(\xi, \zeta) &= \int_{-\frac{1}{2}}^{\frac{1}{2}} \int_{-a(x)}^{a(x)} e^{i(x\xi + y\zeta)} dy dx \\ &= 4 \int_0^{\frac{1}{2}} \cos(x\xi) \frac{\sin(a(x)\zeta)}{\zeta} dx \end{aligned}$$

Maximizing Throughput

Because of the symmetry, we only need to optimize in the first quadrant:

$$\text{maximize } 4 \int_0^{\frac{1}{2}} a(x) dx$$

$$\begin{aligned} \text{subject to } & -10^{-5}E(0,0) \leq E(\xi, \zeta) \leq 10^{-5}E(0,0), & \text{for } (\xi, \zeta) \in \mathcal{O} \\ & 0 \leq a(x) \leq 1/2, & \text{for } 0 \leq x \leq 1/2 \end{aligned}$$


The objective function is the total open area of the mask. The first constraint guarantees 10^{-10} light intensity throughout a desired region of the focal plane, and the remaining constraint ensures that the mask is really a mask.

If the set \mathcal{O} is a subset of the x -axis, then the problem is an infinite dimensional linear programming problem.


One Pupil w/ On-Axis Constraints

$$\rho_{iwa} = 4 \quad \mathcal{T}_{\text{Airy}} = 43\%$$


Small dark zone...Many rotations required



PSF for Single Prolate Spheroidal Pupil



Multiple Pupil Mask



$$\rho_{iwa} = 4$$


$$\mathcal{T}_{\text{Airy}} = 30\%$$

Throughput relative to ellipse

11% central obstr.

Easy to make

Only a few rotations



Space Occulter Design

for Planet-Finding

Space-based Occulter (TPF-O)





Telescope Aperture: 4m, Occulter Diameter: 50m, Occulter Distance: 72,000km


Plain External Occulter (Doesn't Work!)

Shadow \Rightarrow

Circular Occulter




\Leftarrow Note bright spot at center
(Poisson's spot)



Telescope Image



\Leftarrow Shadow (Log Stretch)





Shaped Occulter—Eliminates Poisson's Spot

Shadow \Rightarrow

Shaped Occulter



\Leftarrow Bright spot is gone




\Uparrow
Telescope image shows planet

\Leftarrow Shadow is dark
(Log Stretch)

Apodized Occulters



Apodized Occulter



Radial Attenuation $A(r)$

- The problem (as before) is *diffraction*.
- Abrupt edges create unwanted diffraction.
- *Solution*: Soften the edges with a partially transmitting material—an *apodizer*.
- Let $A(r, \theta)$ denote *attenuation* at location (r, θ) on the occulter.
- The *intensity* of the downstream light is given by the *square of the magnitude of the electric field* $E(\rho, \phi)$.
- *Babinet's principle* plus *Fresnel propagation* gives a formula for the downstream electric field:

$$E(\rho, \phi) = 1 - \frac{1}{i\lambda z} \int_0^\infty \int_0^{2\pi} e^{\frac{i\pi}{\lambda z}(r^2 + \rho^2 - 2r\rho \cos(\theta - \phi))} A(r, \theta) r d\theta dr.$$

where

- z is distance “downstream” and
- λ is wavelength of light.

Attenuation Profile Optimization

$$\begin{array}{ll} \text{minimize} & \gamma \\ \text{subject to} & -\gamma \leq \Re(E(\rho)) \leq \gamma \quad \text{for } \rho \in \mathcal{R}, \quad \lambda \in \Lambda \\ & -\gamma \leq \Im(E(\rho)) \leq \gamma \quad \text{for } \rho \in \mathcal{R}, \quad \lambda \in \Lambda \\ & A'(r) \leq 0 \quad \text{for } 0 \leq r \leq R \\ & -d \leq A''(r) \leq d \quad \text{for } 0 \leq r \leq R \end{array}$$

Specific choice:

$$R = 25, \quad d = 0.04, \quad \mathcal{R} = [0, 3], \quad \Lambda = [0.4, 1.1] \times 10^{-6}$$


where all metric quantities are in meters.

An infinite dimensional linear programming problem.


Discretize:

- $[0, R]$ into 5000 evenly space points.
- \mathcal{R} into 150 evenly spaced points.
- Λ into increments of 0.1×10^{-6} .

Petal-Shaped Occulters



16-Petal Occulter $A(r, \theta)$



Radial Attenuation $A(r)$

- From Jacobi-Anger expansion we get:

$$E(\rho, \phi) = 1 - \frac{2\pi}{i\lambda z} \int_0^R e^{\frac{i\pi}{\lambda z}(r^2 + \rho^2)} J_0\left(\frac{2\pi r \rho}{\lambda z}\right) A(r) r dr \\ - \sum_{k=1}^{\infty} \frac{2\pi(-1)^k}{i\lambda z} \left(\int_0^R e^{\frac{i\pi}{\lambda z}(r^2 + \rho^2)} J_{kN}\left(\frac{2\pi r \rho}{\lambda z}\right) \frac{\sin(\pi k A(r))}{\pi k} r dr \right) \\ \times \left(2 \cos(kN(\phi - \frac{\pi}{2})) \right)$$

where N is the number of petals.

- For small ρ , truncated summation well-approximates full sum.
- Truncated after 10 terms.
- $\lambda \in [0.4, 1.1]$ microns.
- $z = 72,000$ km, $R = 25$ m.
- In angular terms, $R/z = 73$ mas.

Celestial Mechanics: The N -Body Problem

Least Action Principle

Given: n bodies.

Let:

m_j denote the mass and
 $z_j(t)$ denote the position in $\mathbb{R}^2 = \mathbb{C}$ of body j at time t .

Action Functional:

$$A = \int_0^{2\pi} \left(\sum_j \frac{m_j}{2} \|\dot{z}_j\|^2 + \sum_{j,k:k < j} \frac{m_j m_k}{\|z_j - z_k\|} \right) dt.$$

Critical points satisfy equations of motion.

Minimize!

Equations of Motion

First Variation:

$$\begin{aligned}\delta A &= \int_0^{2\pi} \sum_{\alpha} \left(\sum_j m_j \dot{z}_j^{\alpha} \delta z_j^{\alpha} - \sum_{j,k:k < j} m_j m_k \frac{(z_j^{\alpha} - z_k^{\alpha})(\delta z_j^{\alpha} - \delta z_k^{\alpha})}{\|z_j - z_k\|^3} \right) dt \\ &= - \int_0^{2\pi} \sum_j \sum_{\alpha} \left(m_j \ddot{z}_j^{\alpha} + \sum_{k:k \neq j} m_j m_k \frac{z_j^{\alpha} - z_k^{\alpha}}{\|z_j - z_k\|^3} \right) \delta z_j^{\alpha} dt\end{aligned}$$

Setting first variation to zero, we get:

$$m_j \ddot{z}_j^{\alpha} = - \sum_{k:k \neq j} m_j m_k \frac{z_j^{\alpha} - z_k^{\alpha}}{\|z_j - z_k\|^3}, \quad j = 1, 2, \dots, n, \quad \alpha = 1, 2$$

Note: If $m_j = 0$ for some j , then the first order optimality condition reduces to $0 = 0$, which is *not* the equation of motion for a massless body.

The AMPL Model. Nonconvex!!

```
param N := 3; # number of masses
param n := 15; # number of terms in Fourier series representation
param m := 100; # number of terms in numerical approx to integral

param theta {j in 0..m-1} := j*2*pi/m;

var x {i in 0..N-1, j in 0..m-1};
var y {i in 0..N-1, j in 0..m-1};

var xdot {i in 0..N-1, j in 0..m-1}
  = if (j<m-1) then (x[i,j+1]-x[i,j])*m/(2*pi) else (x[i,0]-x[i,m-1])*m/(2*pi);
var ydot {i in 0..N-1, j in 0..m-1}
  = if (j<m-1) then (y[i,j+1]-y[i,j])*m/(2*pi) else (y[i,0]-y[i,m-1])*m/(2*pi);


var KineticEnergy {j in 0..m-1} = 0.5*sum {i in 0..N-1} (xdot[i,j]^2 + ydot[i,j]^2);

var PotentialEnergy {j in 0..m-1}
  = - sum {i in 0..N-1, ii in 0..N-1: ii>i}
      1/sqrt((x[i,j]-x[ii,j])^2 + (y[i,j]-y[ii,j])^2);

minimize A: (2*pi/m)*sum {j in 0..m-1} (KineticEnergy[j] - PotentialEnergy[j]);
```

Choreographies and the Ducati

The previous AMPL model was used to find many *choreographies* (a la Moore and Montgomery/Chencinier) in the equimass n -body problem and the stable *Ducati* solution to the 3-body problem.



<http://www.princeton.edu/~rvdb/JAVA/astro/galaxy/Galaxy.html>